

L'anémomètre à fil chaud : simulation comportementale de son fonctionnement.

J'invite le lecteur à consulter le site pour des informations complémentaires.

Page d'accueil du site Internet :

[page d'accueil](#)

d'autres pdf, sur différents sujets :


[liste des PDF](#)

Parmi les différents instruments de mesure de vitesse du vent, il existe l'anémomètre à fil chaud. Ce document montre le principe du « CTA », (constant temperature anemometer) qui nécessite un asservissement. On étudie dans ce document l'aspect électronique de cet asservissement.

1) Constitution et principe de l'anémomètre à fil chaud

1.a Présentation qualitative

Le capteur est formé d'un **fil** très fin (quelques μm de diamètre), de quelques millimètres de long, chauffé électriquement à une haute température. Le matériau utilisé pour ce fil est à coefficient de température positif : si la température augmente, la résistance électrique augmente.



Ce fil, placé à l'extrémité de 2 longues électrodes, est parcouru par un courant. (Cette sonde ainsi formée est très fragile). Pour mesurer la vitesse du vent, on place ce fil fin face au flux.

Dès cet instant, le vent, à vitesse V_0 par rapport au fil, refroidit ce dernier par convection forcée, et la résistance électrique du fil diminue.


Une électronique d'asservissement rétablit la situation d'origine par une augmentation de l'alimentation de ce fil, pour retrouver la **même résistance et donc la même température du fil**. L'apport d'énergie qu'a nécessité le retour à la température de ce fil donne une indication sur la vitesse du vent.

Hypothèses : 1 - on suppose le vent à la température ambiante. Dans le cas contraire, cela modifierait le refroidissement du fil, et il faudrait en tenir compte, par une correction,
2 - on suppose négligeable l'évacuation de chaleur par conduction avec les électrodes, ainsi que le rayonnement.

1.b. Schéma de principe de l'électronique de mise en œuvre.

Le fil (ou filament) est placé dans un pont de Wheatstone.

R_2, R_3 forment la branche de référence, R_1, R_W forment la branche de mesure.



Le système bouclé cherche à réaliser : $e^- = e^+$.

Avant de placer la sonde sous le vent, on a l'équilibre : $V_s R_3 / (R_2 + R_3) = V_s R_W / (R_1 + R_W)$,
Soit : $R_W = R_1 R_3 / R_2$.

On place le fil chaud sous le vent. R_W diminue et passe à R'_W .

Mais le système bouclé agit pour que e^- suive e^+ . La tension de sortie augmente et passe à $V's$.

On a alors : $V's R_3 / (R_2 + R_3) = V's R'_W / (R_1 + R'_W)$

D'où : $R'_W = R_1 R_3 / R_2$. Les résistances R_1, R_2, R_3 étant inchangées, on a : $R'_W = R_W$.

L'asservissement agit en augmentant V_s , pour replacer la même valeur numérique à la résistance du fil, par réchauffement de ce dernier.


Remarque :

Pour régler le point de fonctionnement, on ajoute un décalage de tension.

L'amplificateur cherche alors à réaliser $e^- = e^+ + V_{offset}$

Soit : $V's R'_w / (R_1 + R'_w) = V's R_3 / (R_2+R_3) + V_{offset}$.

c'est à dire : $R'_w / (R_1 + R'_w) = R_3 / (R_2+R_3) + V_{offset} / V's$.



Cet offset a pour conséquence que le pont diviseur R_1, R_w , est différent de celui formé par R_2, R_3 . Le pont de Wheatstone n'est pas équilibré.

1.c. Mise en équation

On dispose des équations suivantes :


La résistance du fil est une fonction de la température, du type CTP :

$R_{w(T)} = R_0 (1 + \alpha \Delta T)$, (en Ω) où :

α est le coefficient de température, caractéristique du matériau, (en $\Omega/^\circ$)

R_0 valeur à la température ambiante, que l'on pose à T_A ,

ΔT l'élévation de température par rapport à l'ambiante soit $\Delta T = T_w - T_A$.



Puissance dissipée par le fil, loi de Joule : $U_{RW} I_{RW}$ (en W)

où I_{RW} est le courant passant dans le fil chaud, U_{RW} la tension aux bornes du fil chaud.

Par conséquence des hypothèses précédentes, le vent dissipe la puissance délivrée par effet Joule.

On peut modéliser le filament par un petit cylindre dans un courant de fluide. L'étude de la convection de sa chaleur peut se formaliser par la puissance $P = k S \Delta T$,

où k : coefficient d'échange thermique, S : surface d'échange, $\Delta T = T_w - T_A$.

L'expression de k est très complexe, et non traitée dans ce document. (Elle est liée à la conductance thermique du fluide, au diamètre du fil, au nombre de Nusselt, lui-même fonction de divers paramètres).

Une formulation empirique est présentée sous la **formule de King** : $k = a + b \sqrt{v}$.

Où a et b sont des constantes pour un fluide (ici de l'air) et une géométrie du fil donnés, v la vitesse du vent.

Placée dans l'expression précédente, on déduit : $P = (A + B \sqrt{v}) \Delta T$.

On trouve un autre modèle, plus adaptable à la pratique : $P = (A + B v^n) \Delta T$, avec $0,4 < n < 0,6$.

Les valeurs numériques de A et B , issues d'une loi empirique, ne sont accessibles que par un essai de calibration, sur l'anémomètre fonctionnel.

Il vient donc : $U_{RW} I_{RW} = (A + B \sqrt{v}) \Delta T$. On peut isoler ΔT pour le placer dans $R_{w(T)} = R_0 (1 + \alpha \Delta T)$.

Ces équations peuvent se modéliser par le schéma fonctionnel suivant :




Schéma fonctionnel montrant le jeu d'équations gérant $R_w(T)$

Explications :

Le fil chaud, placé entre V(3) et 0 V, est simulé par une **résistance commandée par une tension**, cette dernière représentant ΔT. On applique la relation : $R_{W(T)} = R_0 (1 + \alpha \Delta T)$.

Le courant traversant le fil chaud commande une source de tension V(8), avec un coefficient unitaire (1 A donne 1 V). Un multiplieur réalise alors le produit de 2 tensions, pour délivrer, en V(6), $P = U_{RW} I_{RW}$.

L'offset en amont de l'amplificateur opérationnel positionne un point de fonctionnement à $T = T_A$.

Le vent (variable v) est modélisé par une source de tension placée en V(1). La loi de King est introduite dans le schéma par le biais d'une source de tension commandée par une tension. On a : $V(7) = A + B \sqrt{v}$.

A partir de $P = U_{RW} I_{RW}$, et de $A + B \sqrt{v}$, on réalise la division par une source de tension commandée par une tension. On a ainsi ΔT que l'on injecte dans le modèle de la sonde, par la tension V(5), où 1 V = 1 degré.



La vitesse du vent est donnée, au travers une fonction non linéaire, par V(S). Par une loi de conversion, on peut ensuite afficher la vitesse en m/s.

2) Simulations Pspice

On réalise tout d'abord des tests préliminaires :

2.a Test de la loi de King (fichier test_king.cir)

On simule le schéma suivant. Nous allons adopter les valeurs numériques suivantes **A = 0,003** et **B = 0,0005**.



Simulation de l'équation de la loi de King

Interprétation :

On peut vérifier sur 2 points la relation $A + B\sqrt{v}$:

Pour $v = 0$ V, soit 0 m/s, on a la sortie qui vaut 0,003, soit la valeur de A,

Pour $v = 9$ V, soit 9 m/s, on a la sortie qui vaut $A + B\sqrt{v} = 0,003 + 0,0015 = 0,0045$.

Remarque : V(1) = 0 signifie vent nul, c'est à dire fil placé en dehors du flux que l'on veut mesurer. La loi de King donne alors : $P = A \Delta T$.

C'est une autre formulation de la loi d'Ohm thermique : $\Delta T = R_{TH} P$, où R_{TH} est la résistance thermique (en °/W) entre le fil chaud et l'air ambiant, et P est le flux, la puissance « circulant » dans la résistance R_{TH} , qui provoque un écart de température ΔT.

$\Delta T = R_{TH} P$ avec ΔT en °C, R_{TH} en °/W, P en W.



Par identification, on trouve que $A = 1/ R_{TH}$.

2.b. Test de R_w seul (fichier test_fil.cir)

Le filament à une longueur $l = 2 \text{ mm}$. Son diamètre est $5,5 \text{ }\mu\text{m}$, soit une section $s = 23,76 \cdot 10^{-12} \text{ m}^2$. Il est en tungstène, de résistivité $\rho = 0,55 \cdot 10^{-7} \text{ }\Omega\text{m}$.

Par $R_0 = \rho \frac{l}{s} = 0,55 \cdot 10^{-7} \frac{2 \cdot 10^{-3}}{23,76 \cdot 10^{-12}}$ on obtient $4,6 \text{ }\Omega$.



Le tungstène est donné pour un coefficient $\alpha = 4,5 \cdot 10^{-3}$. L'unité est en K^{-1} ou $^{\circ}\text{C}^{-1}$ selon.

Pour plus de facilité, $R_w(T)$ est intégré dans un **sous-circuit**.

Pour le tester, on place R_w en série avec R_1 (arbitrairement $20 \text{ }\Omega$), alimentée par une tension fixe (arbitrairement 10 V), afin de faire circuler un courant.

Pour vérifier l'équation $R_w(T) = R_0 (1 + \alpha \Delta T)$, on applique une tension de commande en $V(5)$ représentative de ΔT .

La résistance $R_w(T)$, donc $R_w(V(5))$ est obtenue par simple loi d'Ohm : $V(3)/I(R1)$, ou $V(3)/I(Xfilchaud :1)$.



Résistance commandée par une tension

Interprétation :

Pour un ΔT de 100 ° , (par exemple de 20 ° à 120 °), R_w passe de $4,6$ à $4,6 (1 + 0,0045 \times 100) = 6,67 \text{ }\Omega$.

Pour un ΔT de 200 ° , (par exemple de 20 ° à 220 °), R_w passe de $4,6$ à $4,6 (1 + 0,0045 \times 200) = 8,74 \text{ }\Omega$.

Si le filament est maintenu à $120 \text{ }^{\circ}\text{C}$, on a $R_w(120) / R_w(20) = 6,67/4,6 = 1,45$.

De même, à $220 \text{ }^{\circ}\text{C}$, ce rapport vaut $1,9$.


Ce rapport, qui est $(1 + \alpha \Delta T)$, s'appelle le « coefficient de surchauffe ».

Pour bénéficier d'une grande dynamique de mesure de vitesse de vent, il est préférable d'avoir une valeur élevé à ce coefficient, dans les limites technologiques, tout en surveillant que le fil ne rayonne pas ce qui est à craindre avec un faible vent. Dans ce cas, les hypothèses ne seraient plus vérifiées.

2.c. Test global statique (fichier anemometre.cir)

Remarques :

> Sous Pspice, pour accéder, en cours de simulation, à un courant dans une branche, il est nécessaire de placer une source de tension nulle. Cette dernière a été ajoutée en série à R_w afin de disposer de I , pour la source « H » (tension commandée par un courant) qui délivre $V(8)$.




ajout d'une source de tension nulle

> Il s'agit d'une simulation comportementale :

- Les opérateurs réalisant la multiplication, division, sont parfaits.
- On suppose que les autres composants (résistances, A. Op.) sont insensibles à la température.
- En pratique, le courant débité dans la branche R_1, R_w peut nécessiter un transistor de puissance, monté en émetteur suiveur, en sortie de l'A. Op.
- On néglige la résistance des fils de liaison de la sonde.

- L'amplificateur a un gain statique de 106 dB (amplification A_0 de 200 000) et une fréquence de coupure de 5 Hz. La fréquence de transition est de 1 MHz.


Ce sont les caractéristiques d'un LM741.



$R_1 = 4 \Omega$ $R_2 = 150 \Omega$ $R_3 = 125 \Omega$ $V_{offset} = 400 \text{ mV}$

- On vérifie la dynamique de fonctionnement en réalisant un balayage de $V(1)$ de 0 V à 10 V. Cela correspondrait à placer la sonde dans une soufflerie dont le vent est réglé pour aller, de façon très lente, de 0 à 10 m/s (= 36 km/h).

→ Regardons la **tension de sortie** :



V(S), tension de sortie selon la vitesse du vent.

Pour $V(1) = 0$, soit vent nul, on a la tension de sortie qui vaut 2,3083 V.
 Pour $V(1) = 10$, soit $v = 10 \text{ m/s}$, on a la tension de sortie qui vaut 2,5334 V.
 Dans cet essai, $V(S)$ a augmenté de 0,225 V.

Equation $V(S) = f(v)$:

Cette équation peut se déterminer graphiquement. On lui adopte le modèle $V(S) = C + D v^n$.

➤ La valeur de C est immédiate : par $v = 0$, on a $C = 2,3083$.
 On peut en déduire $V(S) - 2,3083 = D v^n$.

En prenant le log de cette expression : $\log [V(S)-2,3083] = \log D + n \log v$.

➤ Le log D n'étant qu'une translation, la détermination de n s'obtient par une mesure de pente de cette fonction :

On affiche la courbe $\log_{10}(V(S)-2,3083)$ en ordonnée et $\log_{10}(V_{vent})$ en abscisse, et on mesure entre 2 points, comme -1 et 1.

Trace Name	Y1	Y2	Y1 - Y2
X Values	-1.0000	1.0000	-2.0000
LOG10(V(S)-2.3083)	-1.5786	-647.727m	-930.828m

$930,828\text{m} / 2 = n = 0,465$.

➤ La valeur de D s'identifie alors pour v quelconque, par exemple $v = 10 \text{ m/s}$:
 On a $2,5334 = 2,3083 + D 10^{0,465}$, d'où $D = (2,5334 - 2,3083) / 10^{0,465} = 0,2251 / 2,917 = 0,077$.

On peut donc donner l'équation : $V(S) = 2,3083 + 0,077 v^{0,465}$.


Pour vérifier, il suffit, sous Probe, d'afficher la courbe « $2.3083+0.077*\text{PWR}(V(1),0.465)$ » qui se superpose parfaitement à $V(S)$.

Remarque :

La relation entre $V^2(S)$ et le vent admet aussi le même modèle.

Par une démarche absolument similaire (tracé de $V(S)*V(S)$ en fonction de v, puis passage par les log) on aboutit à : $V^2(S) = 5,3284 + 0,365 v^{0,4748}$.


→ Observons la valeur de R_w :



Valeur de R_w selon la vitesse du vent.

Elle est accessible par application de la loi d'Ohm : il suffit d'afficher $V(3)/I(Xfilchaud:1)$. Sur toute la dynamique de vent (de 0 à 10 m/s), la valeur de la résistance R_w faiblit peu : de **6,7477 Ω** à **6,3208 Ω** , soit une variation de 6,3 %. L'asservissement de résistance présente une faible erreur. Remarque : à $v = 5$ m/s, $R_w = 6,4208 \Omega$. Cette valeur sera utilisée dans l'annexe 1.

→ Si on affiche $V(5)$, on dispose de ΔT :



ΔT selon la vitesse du vent.


Pour $V(1) = 0$, soit vent nul, on a 103,75 V.

Ce qui signifie que le fil s'est échauffé de 103,75 °C par la loi d'Ohm thermique. Si $T_A = 20$ °C, le fil chaud est à 123,75 °C.

Pour $V(1) = 10$, soit $v = 10$ m/s, on a 83,131 V, donc $\Delta T = 83$ °C.

Ce qui signifie que le fil est à 103 °C. Malgré l'asservissement, le fil a perdu 20,6 °C par la convection forcée créée par le vent.

→ Une fonction de transfert intéressante : $V(S) = f(\Delta T)$



$V(S) = f(\Delta T)$

Dans toute la dynamique, cette caractéristique est linéaire. La pente est $-0,011$ V / °C.

On retrouve que l'asservissement de température présente une erreur : malgré l'augmentation de V_s de 0,225 V, le fil chaud n'a pas conservé sa température et a perdu 20,6 °C.

2.d. Test global dynamique

Rappelons qu'en signaux variables, il y a lieu de considérer $i(t)$, $v(t)$ et la puissance $p(t) = u(t) i(t)$. Dans le cas où $u(t)$ et $i(t)$ sont constants (en continu), on retrouve la puissance par $P = U I$.

Capacité thermique du fil

Bien que de petite taille, le fil dispose d'une capacité thermique. Dans notre article, elle peut être évaluée à $1,1 \cdot 10^{-7} \text{ J/}^\circ$, au voisinage du point de fonctionnement (à $v = 5 \text{ m/s}$). Associée à la résistance thermique, dont la valeur est $R_{TH} = 1 / A = 333 \text{ }^\circ/\text{W}$, cela forme une constante de temps d'environ $37 \mu\text{s}$. On peut introduire cette constante de temps en plaçant en parallèle avec R_W un condensateur C_{TH} modélisant cette capacité thermique.

Pour respecter $R_W C_{TH} = 37 \mu\text{s}$, on trouve $C_{TH} = 37 \mu\text{s} / 6,4208 \Omega = 5,76 \mu\text{F}$.

Circuit correcteur

Les premières simulations ont montré que la présence de C_{TH} déstabilisait le montage. Le système est formé de 2 boucles, comme expliqué en annexe 1, qui montre que l'apport du condensateur $C_1 = 6,8 \mu\text{F}$ placé en parallèle à R_1 permet de stabiliser le système.





Schéma complet

Pour tester l'anémomètre en simulation, on peut placer un « échelon de vent », c'est à dire un vent qui évolue instantanément entre 2 valeurs, comme de 0 m/s à 10 m/s. La transition instantanée de l'échelon est purement théorique. On suppose que le régime transitoire est très bref pour qu'on puisse raisonner sur l'équivalent de 2 essais continus.



Courbe supérieure : échelon de vent de 0 à 10 m / s,
 courbe intermédiaire : valeur de la résistance R_W ,
 courbe inférieure : $V(S)$, tension de sortie.

Explications :

- Ce n'est pas simulé ici, mais, hors tension, le fil est à T_A , température ambiante, donc $R_W = R_0$.

- Sous tension, sonde hors vent : c'est l'intervalle **entre 0 et 0,2 ms**.

La résistance du fil chaud augmente par son élévation de température. Le système bouclé agit pour placer le point de repos tel que $V(3) = V(2) + V_{\text{offset}}$, soit, en affichant ces potentiels : $V(3) = 1,45 \text{ V}$.

C'est donc les valeurs des résistances du pont de Wheatstone, associé à V_{offset} , qui gèrent $V(3)$, donc R_W , qui vaut **6,7477 Ω** . La tension de sortie vaut 2,3083 V. Ces 2 valeurs ont déjà été mesurées au début de la courbe issue de l'essai statique. Le coefficient de surchauffe est $R_W(120) / R_W(20) = 6,74/4,6 = 1,46$.

Par loi de Joule, le fil chaud développe une certaine puissance P dissipée par convection naturelle dans l'air ambiant. Le potentiel $V(6)$ indique 0,311 W.

$V(7)$ prend la valeur $A = 0,003$. La loi d'ohm thermique donne $\Delta T = P/A$ soit $0,311 / 0,003 = 103,7 \text{ }^\circ\text{C}$, valeur indiquée par $V(5)$. Comme indiqué dans l'essai statique : si la température ambiante = 20 ° , alors le fil chaud est à $123,7 \text{ }^\circ\text{C}$.

- On place le fil chaud sous le flux d'air à mesurer : c'est l'intervalle **entre 0,2 ms et 0,7 ms**.

$V(S)$ augmente à 2,5334 V, pour assurer une nouvelle égalité $V(3) = V(2) + V_{\text{offset}}$, soit, en affichant ces potentiels : $V(3) = 1,55 \text{ V}$. R_W est passée à **6,3208 Ω** en régime établi. Ces 2 valeurs ont déjà été mesurées lors de l'essai statique. La puissance est passée à 0,380 W. ΔT ne vaut plus que $83,11 \text{ }^\circ\text{C}$.

- On remarque très nettement un régime transitoire sur $R_W(t)$. Lors de l'installation du vent, la résistance de fil chaud diminue instantanément. Mais l'asservissement agit pour replacer l'ancienne valeur à R_W , ce qui explique le pic vers le bas. Comme signalé, une erreur de position est présente, puisqu'on ne retrouve pas exactement l'ancienne valeur. La transition n'est pas instantanée : le temps de réponse est d'environ 40 μs . C'est essentiellement lié à la présence de C_1 , dont le rôle est indiqué en annexe, qui détermine ce temps.

- on retire instantanément le fil chaud du flux d'air: c'est l'intervalle **entre 0,7 ms et 1 ms**.

C'est le retour à la situation de départ.


- On peut estimer la puissance consommée par cet anémomètre : soit en additionnant les différentes puissances (0,380 W dans le fil + 0,241 W dans $R_1 + \dots$) soit en affichant celle débitée par l'A. Op. : on a, durant la phase active, 645 mW. Cet appareil peut être alimenté par une pile de 4,5 V.

Linéarisation

Pour faciliter la lecture de la vitesse du vent, le montage est suivi d'une fonction qui permet d'avoir une relation linéaire entre l'entrée, la vitesse du vent, et la sortie, l'affichage de la valeur.

Dans notre simulation, cet étage est modélisé par une source de tension commandée en tension, dont l'équation est la réciproque de $V(S) = f(v)$.

Cette équation peut s'extraire de la caractéristique DC : $V(1)$ en ordonnée et $V(S)$ en abscisse :




Courbe de linéarisation

Par exportation sous Excel, on peut déterminer une équation pour cette fonction, de degré 2, ou, pour avoir un coefficient de détermination $R^2 = 1$, de degré 3 : **$338,78 x^3 - 2231,6 x^2 - 4888,6 x + 3560,6$** .

Cette équation permet de « remonter » à la vitesse du vent à partir de la tension de sortie. Cette linéarisation est la dernière partie de l'anémomètre. Dans la simulation, on exploite une nouvelle source de tension commandée par une tension pour délivrer le potentiel $V(10)$: 1 V représente 1 m/s, qui, convertie en numérique, sera affichée sur des digits.

Le test final consiste à présenter un profil de vent quelconque en $V(1)$, et de visualiser en $V(10)$: L'échelle des temps est arbitraire : on représente ici 10 ms, (pour éviter un run trop long), avec paliers, montées, descente...



Le montage complet donne une information parfaitement similaire à la vitesse du vent

Interprétation :

En sortie du linéarisateur, on a une représentation fidèle de la vitesse du vent, avec un facteur d'échelle unitaire : 1 m/s de vent donne 1 V de tension à afficher.

CONCLUSION

Cet article a montré le principe de l'anémomètre à fil chaud, et, grâce aux simulations, a donné quelques renseignements réalistes : alimentation sous 4,5 V, consommation 650 mW (hors affichage), gamme de mesure de 0 à 10 m/s, temps de réponse environ 40 μ s.



Dans un but de simplification, on n'a pas simulé l'impédance du câble connecté au fil chaud, ni abordé le schéma de l'amplificateur à base de 2 étages, couramment utilisé en pratique.

L'annexe 1 ci-après donne une représentation « système bouclé », pour étudier de la stabilité du montage complet.


Annexe 1 : étude de la stabilité

1) Exploitions tout d'abord le **schéma de principe** :

L'électronique d'asservissement montre un amplificateur rebouclé, que l'on peut représenter par le formalisme des schémas blocs :



Si on suppose R_w constant, le système est linéaire et on peut procéder à une transformation du schéma bloc :



La TBO est :

$$-\frac{A}{1 + A \frac{R_w}{R_1 + R_w}} \frac{R_3}{R_2 + R_3}$$


La TBF est : $-\frac{A}{1 + A(\frac{R_w}{R_1 + R_w} - \frac{R_3}{R_2 + R_3})}$. Simplifions son écriture par $-\frac{A}{1 + A\beta}$.

La réponse harmonique de l'A. Op. est de la forme : $\frac{A_0}{1 + j\omega\tau}$.

Sa transformée de Laplace est $A(p) = \frac{A_0}{1 + \tau p}$.


La TBF(p) devient : $-\frac{\frac{A_0}{1 + \tau p}}{1 + \frac{A_0}{1 + \tau p}\beta} = -\frac{A_0}{1 + \tau p + A_0\beta} = -\frac{A_0}{(1 + A_0\beta)(1 + \tau'p)}$, où $\tau' = \frac{\tau}{1 + A_0\beta}$

La stabilité peut alors se traiter par la position du pôle : dans le plan complexe, il faut, pour avoir la stabilité, qu'il se situe dans le demi-plan gauche. Le pôle étant $-1/\tau'$, cela se traduit par $\tau' > 0$, ce qui est obtenu, si $(1 + A_0\beta)$ est > 0 .



Le système sera stable si $A_0\beta > -1$

La condition de stabilité s'écrit également $\beta > -\frac{1}{A_0}$



Si $A \rightarrow \infty$ (cas de notre article), $1 / A_0 \rightarrow 0$. Le système sera stable si $\beta > 0$ soit $\frac{R_w}{R_1 + R_w} > \frac{R_3}{R_2 + R_3}$.

Cette inégalité pouvait se déduire intuitivement : la contre réaction doit prédominer la réaction positive.

Vérifions cela, par exemple à $v = 5 \text{ m/s}$, où $R_w = 6,4208 \Omega$.

Dans le pont de Wheatstone, on place une résistance fixe de $6,4208 \Omega$ au lieu du sous-circuit R_w .

Nous avons $\frac{R_w}{R_1 + R_w} = \frac{6,4208}{4 + 6,4208} = 0,616$ et $\frac{R_3}{R_2 + R_3} = \frac{125}{150 + 125} = 0,4545$: $\beta > 0$ est vérifié.

Un calcul donne le pole placé en $-1,0164 \cdot 10^6$.

A la simulation temporelle, $V(S)$ est stable et vaut $2,4751 \text{ V}$.

Cette stabilité est conservée en diminuant R_2 jusqu'à 77Ω .

En effet, avec R_2 en 77Ω , nous avons $\frac{R_3}{R_2 + R_3} = \frac{125}{77 + 125} = 0,618$: $\beta = -0,002$

En simulation $V(S)$ commence à -150V et diverge après 2 ms .


2) Observons le rôle des condensateurs :

La simulation montre que la présence de ce condensateur C_{TH} déstabilise le montage.
D'où la présence de C_1 qui agit en correcteur.

Expliquons cela, sur le schéma équipé de C_{TH} et C_1 :

Dans les équations précédentes, remplaçons R_W par $\frac{R_W}{1 + R_W C_{TH} p}$

et R_1 par $\frac{R_1}{1 + R_1 C_1 p}$.



Le pont diviseur $\frac{R_W}{R_1 + R_W}$ est remplacé par $\frac{\frac{R_W}{1 + R_W C_{TH} p}}{\frac{R_1}{1 + R_1 C_1 p} + \frac{R_W}{1 + R_W C_{TH} p}} = k \frac{1 + \tau_N p}{1 + \tau_D p}$

où $k = \frac{R_W}{R_1 + R_W}$ $\tau_D = Req Ceq$ avec $Req = R_1 // R_W$ et $Ceq = C_1 // C_{TH}$ $\tau_N = R_1 C_1$

Le terme β devient : $k \frac{1 + \tau_N p}{1 + \tau_D p} - \frac{R_3}{R_2 + R_3} = \frac{k(1 + \tau_N p)(R_2 + R_3) - R_3(1 + \tau_D p)}{(1 + \tau_D p)(R_2 + R_3)}$

L'expression de la TBF(p) : $-\frac{A_0}{1 + \tau p + A_0 \beta}$ devient : $-\frac{A_0}{1 + \tau p + A_0 \left(\frac{k(1 + \tau_N p)(R_2 + R_3) - R_3(1 + \tau_D p)}{(1 + \tau_D p)(R_2 + R_3)} \right)}$

Ce qui donne, après 5 lignes de calcul :

$$TBF(p) = -\frac{A_0(1 + \tau_D p)}{p^2 \tau \tau_D + p \left\{ [(\tau + \tau_D) + A_0 [k \tau_N - \tau_D R_3 / (R_2 + R_3)]] + 1 + A_0 [k - R_3 / (R_2 + R_3)] \right\}}$$

Le système sera stable si tous les coefficients du dénominateur sont de même signe.

Soit : $\tau \tau_D > 0$ (toujours vérifié)

$$(\tau + \tau_D) + A_0 [k \tau_N - \tau_D R_3 / (R_2 + R_3)] > 0$$

$$1 + A_0 [k - R_3 / (R_2 + R_3)] > 0$$

Vérifions ce calcul avec notre application numérique (au voisinage de 5 m/s) :

$A_0 = 200\,000$ $\tau = 31,8 \text{ ms}$ $R_1 = 4 \Omega$ $R_2 = 150 \Omega$ $R_3 = 125 \Omega$ $C_{TH} = 5,76 \mu\text{F}$
 $R_W = 6,4208 \Omega$ $R_3 / (R_2 + R_3) = 0,4545$ $k = R_W / (R_1 + R_W) = 0,616$ $Req = 2,4646 \Omega$

Premier cas $C_1 = 4,7 \mu\text{F}$ $\Rightarrow Ceq = 10,46 \mu\text{F}$ $\tau_D = 25,78 \mu\text{s}$ $\tau_N = 18,8 \mu\text{s}$

$$(\tau + \tau_D) + A_0 [k \tau_N - \tau_D R_3 / (R_2 + R_3)] = 31,826 \cdot 10^{-3} + 200\,000 [11,58 - 11,718] \cdot 10^{-6} = 4,34 \cdot 10^{-3} > 0$$

$$1 + A_0 [k - R_3 / (R_2 + R_3)] = 1 + 200\,000 [0,616 - 0,4545] = 32301 > 0$$

Le système est **stable**

Autre cas $C_1 = 4,5 \mu\text{F}$ $\Rightarrow Ceq = 10,26 \mu\text{F}$ $\tau_D = 25,286 \mu\text{s}$ $\tau_N = 18 \mu\text{s}$

$$(\tau + \tau_D) + A_0 [k \tau_N - \tau_D R_3 / (R_2 + R_3)] = 31,825 \cdot 10^{-3} + 200\,000 [11,088 - 11,4925] \cdot 10^{-6} = -0,049 < 0$$

Le système est **instable**.

On remarque, dans cet essai pont de Wheatstone + A. Op., qu'il faut $C_1 > 4,7 \mu\text{F}$ pour avoir la stabilité.

Annexe2 : fichiers.cir

```

Loi de King
* fichier test_king.cir
Vvent 1 0 DC=0
.param n = 0.5
.param A = 0.003
.param B = 0.0005
Eking 7 0 VALUE = { {A} + {B}*PWR(V(1),{n}) }
.DC Vvent 0 10 10m
.probe
.end

```

```

fil chaud
* fichier test_fil.cir
Valim S 0 DC=10V
R1 S 3 20 ; 20 ohm en série avec RW
Xfilchaud 3 0 1 0 RW
VdeltaT 1 0 DC=0
.param R0 = 4.6 ; valeur de RW à TA
.param alpha = 0.0045 ; coef de température
.SUBCKT RW Rout1 Rout2 in1 in2 ;
ETHERM Rout1 3 VALUE = { I(VSENSE)*{R0}*(1+{alpha}*(V(in1,in2))) }
VSENSE 3 Rout2 DC 0
.ENDS
.DC VdeltaT 0 200 10m
.probe
.end

```

```

pont de Wheatstone et anemometre
* fichier anemometre.cir
R1 S 3 4
R2 S 2 150
R3 2 0 125

.SUBCKT RW Rout1 Rout2 in1 in2 ;
ETHERM Rout1 3 VALUE = { I(VSENSE)*4.6*(1+0.0045 *(V(in1,in2))) }
VSENSE 3 Rout2 DC 0
.ENDS

Xfilchaud 3 20 5 0 RW
Vbidon 20 0 DC=0 ; pour prélever le courant
HU 8 0 POLY(1) Vbidon 0 1 ; délivre U, image de I
EmultP 6 0 POLY(2) (8,0) (3,0) 0 0 0 0 1 ; délivre U I = P

Vvent 1 0 DC=0 PULSE (0 10 0.2m 0.1u 0.1u 0.5m 1m) ; 10 m/s entre 0,2ms et 0,7 ms
Eking 7 0 VALUE = { 0.003 + 0.0005*PWR(V(1),0.5) }

EdeltaT 5 0 VALUE = {V(6)*PWR(V(7),-1) } ; réalise V(6) / V(7)
* ampli
Voffset 4 0 DC=0.4
EAop S 0 LAPLACE {V(2)-V(3)+V(4)} {{200000}/(1+0.0318*s) }
ES 10 0 VALUE = {338.78*PWR(V(S),3) - 2231.6*PWR(V(S),2) + 4888.6*V(S) - 3560.6 }
Cth 3 20 5.76u
C1 S 3 6.8u

.DC Vvent 0 10 1m
.TRAN 1u 1m 0u 1u
.probe
.end

```

articles 1 à 43 : sur le livre

Tableau récapitulatif des articles PDF disponibles sur ce site

n°	titre	lien présentation	lien direct article
	Guide d'installation et d'emploi simplifié	présentation	document PDF
44	Exemples basiques et des exercices...	présentation	document PDF
45	Un exemple de circuit passif	présentation	document PDF
46	Un oscillateur Colpitts	présentation	document PDF
47	Compensation en fréquence des amplificateurs opérationnels	présentation	document PDF
48	Un amplificateur à transistors bipolaires	présentation	document PDF
49	Une bascule D Flip Flop CMOS	présentation	document PDF
50	Une porte XOR à transistors MOS	présentation	document PDF
51	Un VCO à 12 transistors MOS	présentation	document PDF
52	Une PLL à moins de 20 transistors MOS	présentation	document PDF
53	Un oscillateur à résistance négative	présentation	document PDF
54	Une charge électronique	présentation	document PDF
55	Un amplificateur en classe C	présentation	document PDF
56	Le monostable 74 123	présentation	document PDF
57	Un amplificateur en classe D	présentation	document PDF
58	Le transformateur en linéaire	présentation	document PDF
59	La loi d'ohm thermique	présentation	document PDF
60	Le transformateur en non linéaire	présentation	document PDF
61	Robustesse d'un oscillateur en anneau	présentation	document PDF
62	Une alimentation stabilisée	présentation	document PDF
63	Modélisation d'un haut-parleur	présentation	document PDF
64	Un synthétiseur de fréquence	présentation	document PDF
65	Un ampli audio de Sparkfun	présentation	document PDF
66	Simulation logique et analogique	présentation	document PDF
67	Un oscillateur à relaxation	présentation	document PDF
68	Lecteur de TAG RFID 125 kHz	présentation	document PDF
69	Diagramme de l'œil avec Pspice	présentation	document PDF
70	Un amplificateur hautes fréquences	présentation	document PDF
71	Une bizarrerie enfin expliquée...	présentation	document PDF
72	Comprendre le paramétrage de la FFT	présentation	document PDF
73	La relation de Bennett	présentation	document PDF
74	Simuler un circuit à plus de 20 transistors avec PSpice Eval	présentation	document PDF
75	Une horloge biphase sans recouvrement	présentation	document PDF
76	Quelques simulations sur la diode	présentation	document PDF
77	Un ampli classe A, avec transformateur de sortie	présentation	document PDF
78	Des stimuli pour PSpice	présentation	document PDF
79	Simuler le TL431 : zener ajustable	présentation	document PDF
80	Un ADC flash	présentation	document PDF
81	Une chaîne d'acquisition : S&H, ADC, DAC	présentation	document PDF
82	Un amplificateur 50 MHz	présentation	document PDF
83	Un dérivateur non inverseur	présentation	document PDF
84	Un amplificateur bipolaire avec push pull CMOS	présentation	document PDF
85	Rôle des répéteurs logiques dans un circuit intégré	présentation	document PDF
86	Un driver logique CMOS pour charge 50 ohms	présentation	document PDF
87	Des triggers de Schmitt et des applications	présentation	document PDF
88	Un filtre gaussien analogique	présentation	document PDF
89	Un générateur de bruit rose	présentation	document PDF

90	Un anémomètre à fil chaud : simulation comportementale	présentation	document PDF
91	Un oscillateur à pont de Wien stabilisé par CTN	présentation	document PDF
92	L'emballement thermique d'une diode	présentation	document PDF
93	Les puissances dans un amplificateur	présentation	document PDF
94	Asservissement de puissance dans une résistance	présentation	document PDF
95	Asservissement de la puissance émise par une antenne radio	présentation	document PDF
96	Un driver de LED de puissance	présentation	document PDF
97	Exploiter Pspice pour simuler des filtres numériques	présentation	document PDF
98	Un filtre en cosinus surélevé avec Pspice	présentation	document PDF
99	Effet de la température sur un amplificateur en classe A	présentation	document PDF
100	Un amplificateur à transistors JFET et bipolaires	présentation	document PDF
Supplément, hors article :			
mon cours « Electronique pour les communications numériques », polycopié couleur 201 pages en pdf			

[retour à l'écran d'accueil de ce site](#)